

The Maple Package “Janet”: I. Polynomial Systems

Yuri A. Blinkow¹, Carlos F. Cid², Vladimir P. Gerdt³,
Wilhelm Plesken², Daniel Robertz²

¹ Department of Mathematics and Mechanics,
Saratov University, 410071 Saratov, Russia

² Lehrstuhl B für Mathematik, RWTH Aachen,
Templergraben 64, D-52062 Aachen, Germany

³ Laboratory of Information Technologies, Joint Institute for Nuclear Research,
141980 Dubna, Russia

Abstract

The MAPLE package “Janet” comes in two parts, the first dealing with polynomials and commutative algebra, the second with linear pdes. Here the first part is described. Amongst others it contains a MAPLE and a C++ implementation of the involutive algorithm for polynomial modules as an alternative for conventional Gröbner basis techniques.

1 Introduction

This is the first of two papers introducing the Maple package “Janet”. The package contains the first Maple implementation of the Involutive algorithm [GeB 98a], [GeB 98b] in the form presented in [Ge 02]. More precisely, our implementation fixes JANET separation of variables into multiplicative and non-multiplicative ones [Jan 29] as the input involutive division [GeB 98a] for the algorithm that produces a polynomial Janet basis in the output. By means of this technique the canonical involutive normal forms for systems of linear partial differential equations can be produced. In this first paper we shall only describe a version of implementation of the algorithm for polynomial systems and various related algorithms dealing with commutative algebra. Theoretically speaking, the polynomial version is obtained from more general differential version [Ge 99] if one restricts to linear systems of pdes with constant coefficients, cf. [Ple 02] for more details. From the point of view of implementation, however, quite some effort has been taken to optimise the polynomial version, so that it competes with implementations of the traditional GRÖBNER basis implementations, cf. [BGe 01] for time comparisons. To make the package usable for bigger systems, C++ routines for the most important functions have been built in. Without the C++ code the package has about 4500 line of Maple code.

Below is a list of the commands available in the polynomial part of “Janet”: With the commands in the first group one can create a JANET basis, produce normal forms of elements of the residue class module, print out the JANET basis with some relevant extra information and get quantitative information about the module as well as an explicit basis, even in the infinite dimensional case. The usage with definitions and some typical examples and comments on the algorithms will be given in section 2, more details on steering the main function is given in the subsequent section 3. The section 4 comments on relations between generators (corresponding to compatibility conditions in the pde case). Again we proceed in the scheme definitions, algorithms, examples. Finally the last section 5 comments on various elimination techniques by giving examples, i. e. computing intersections of modules, finding annihilators, finding ring relations. We have taken big effort to produce detailed online documentation in the form of help pages with examples, so that we can be brief in this account.

Basic commands:	
InvolutiveBasis	PolInvReduce
PolTabVar	PolHilbertSeries
FactorModuleBasis	
Commands for special applications:	
PolMinPoly	Syzygies
PolResolution	PolRepres
Relations	PolWeightedHilbertSeries
NotHas/Has	
Commands for various invariants derivable from PolHilbertSeries:	
PolIndexRegularity	PolDimension
PolHilbertPolynomial	PolHilbertFunction
PolHP	PolHF
PolCartanCharacter	
Auxiliary Commands:	
LeadingMonomial	Stats
AddRhs	

2 Basics

Let $R := K[x_1, \dots, x_n]$ be the polynomial ring over the field K in the indeterminates x_1, \dots, x_n . In general K can be any field, in which one has a normal form for elements and where addition, subtraction, multiplication, and division can constructively be carried out. For our MAPLE implementation it means $K = \mathbb{Q}$ or a finitely generated field extension of \mathbb{Q} . Given q -tuples $A_1, \dots, A_a \in R^q$ two problems are treated:

1. Find a K -basis for the R -module $M := R^q/S$, where S is the submodule of R^q generated by A_1, \dots, A_a .
2. Given $t \in M$, express t in terms of this basis.

Note the special case $q = 1$, where S becomes an ideal and M a residue class ring of R . This is also the case most relevant to polynomial equations. However, the general case is also often relevant in physical situations, e. g. for linear systems of pdes with constant coefficients, where the x_i become partial derivatives or if one use LAPLACE transform stay variables.

The above task is dealt with by the involutive algorithmic technique, which is first applied to create a JANET basis J for M that allows to read off a K -basis for M , and then apply the involutive reduction with respect to the JANET basis to have an algorithm for task 2. The JANET basis B is a finite subset of S which is more than a GRÖBNER basis: it does not only have the property that one can constructively represent any element of S as a linear combination of the elements of B , but the concept of multiplicative variables restricts the coefficients of the linear combinations in such a way that not only the coefficients become unique but also the algorithmic procedure to obtain them leaves no space for a long way round but forces every move by strict rules. Like in the ordinary GRÖBNER basis case the result depends to some extend on an ordering of the monomial K -basis of R^q . Various possibilities are realized, the degree reverse lexicographic option with the additional option of favouring the component with the highest degree is usually chosen. Here is a rough description of in- and output:

Input: $A_1, \dots, A_a \in R^q$ generating the submodule S of R^q (and a list of indeterminants, e. g. x_1, \dots, x_n .)

Output: The JANET basis B_1, \dots, B_d of S by the call `InvolutiveBasis`.

The subsequent call `PolTabVar` reproduces each B_i , the leading term of B_i and the subset $M_i \subseteq \{x_1, \dots, x_n\}$ of multiplicative variables of B_i with respect to B , i. e. each element s of S has a unique representation as

$$s = \sum_{i=1}^d p_i B_i \text{ with } p_i \in K[x_i | x_i \in M_i].$$

The further subsequent call `FactorModuleBasis` produces a subset of all monomial K -basis vectors $(0, \dots, 0, x_1^{\alpha_1} \dots x_n^{\alpha_n}, 0, \dots, 0)$ (with degree $\alpha_1 + \dots + \alpha_n$) of R^q whose residue classes modulo S form a K -basis of $M = R^q/S$.

The command `PolHilbertSeries` gives the generating function for the numbers of these basis vectors according to their degrees.

Finally, with further **input** $v \in R^q$ the command `PolInvReduce` produces the normalized representative of the coset $v + S \in M$ i. e. the unique K -linear combination \bar{v} of the basis vectors above with $\bar{v} + S = v + S$.

Example 2.1 $q = 1$, infinite dimensional residue class ring:

Specification of variables:

> `var := [x, y, z];`

`var := [x, y, z]`

Ideal generators:

> `L := [x^2+y^2-1, x^2+z^2-2];`

$$L := [x^2 + y^2 - 1, x^2 + z^2 - 2]$$

Computation of Janet basis:

> $J := \text{InvolutiveBasis}(L, \text{var});$

$$J := [-z^2 + 1 + y^2, x^2 + z^2 - 2, -z^2 x + x + y^2 x]$$

Janet basis with multiplicative variables and leading terms:

> $\text{PolTabVar}();$

$$[-z^2 + 1 + y^2, [* , 2, 3], y^2]$$

$$[x^2 + z^2 - 2, [1, 2, 3], x^2]$$

$$[-z^2 x + x + y^2 x, [* , 2, 3], y^2 x]$$

The expansion via the geometric series of the following yields the sum of the monomials forming a K -basis for the vector space complement of the ideal in the polynomial ring:

> $F := \text{FactorModuleBasis}(\text{var});$

$$F := \frac{x}{1-z} + \frac{yx}{1-z} + \frac{1}{1-z} + \frac{y}{1-z}$$

The count for the basis vectors according to degree: (Note, since $(1-t)$ is the highest power of $(1-t)$ occurring in the denominators, the (maximal) dimension of the variety is 1.)

> $\text{PolHilbertSeries}(t);$

$$1 + 3t + 4t^2 + 4\frac{t^3}{1-t}$$

Computing the normalized representative of the coset of x^3 :

> $\text{PolInvReduce}(x^3, J, \text{var});$

$$-z^2 x + 2x$$

Example 2.2 $q = 2$, finite dimensional residue class module:

> $\text{var} := [x, y];$

$$\text{var} := [x, y]$$

> $L := [[x, -y], [y, x], [y^3, 0]];$

$$L := [[x, -y], [y, x], [y^3, 0]]$$

> $J := \text{InvolutiveBasis}(L, \text{var});$

$$J := [[y, x], [x, -y], [0, x^2 + y^2], [0, y^2 x], [0, y^4]]$$

> $\text{PolTabVar}();$

$$[[y, x], [* , 2], [y, 1]]$$

$$[[x, -y], [1, 2], [x, 1]]$$

$$[[0, x^2 + y^2], [1, 2], [x^2, 2]]$$

$$[[0, y^2 x], [* , 2], [y^2 x, 2]]$$

$$[[0, y^4], [* , 2], [y^4, 2]]$$

> $\text{FactorModuleBasis}(\text{var});$

$$[[0, 1], [0, y], [0, x], [0, y^2], [0, yx], [0, y^3], [1, 0]]$$

```

> PolHilbertSeries(t);
                2 + 2t + 2t2 + t3
> PolInvReduce([x2, y2], J, var);
                [0, yx + y2]

```

3 Modifications and Extensions

There are quite a few possibilities to modify the above computations: In `InvolutiveBasis` various orderings for the monomial basis can be chosen. The standard choice is the degree reverse lexicographical ordering for the monomials itself. A second possibility is the pure lexicographic ordering, which might be slow for big examples, but is often used for elimination purposes. In both cases the ordering of the variables is implicitly declared by the declaration of the variables. In the proper module case, i. e. $q > 1$, one can in addition choose the ordering of the tuples. The default strategy is “term over position”, i. e. the leading term of a tuple is the highest among the leading terms of the components. In case it is not unique the first one is chosen. The strategy “position over term”, where the leading term is the leading term of the first non zero component, is also possible. In any case one can modify the natural succession of the components by a permutation. The default strategy “term over position” is usually much more effective, but the other strategy is sometimes useful, when one computes resolutions. Note, all of these modifications might lead to different JANET bases. If one has not used the defaults, one should also repeat the specification with every call of `PolInvReduce`, to be sure that it terminates. Other changes implied are the multiplicative variables and the grading of the residue class module. This latter point shows up in the commands `FactorModuleBasis` and `PolHilbertSeries`.

Example 3.1 (*Influence of the orderings*):

First we work with the degree reverse lexicographical ordering (note $x > y$):

```

> var := [x, y];
                var := [x, y]
> L := [x - y3];
                L := [x - y3]
> J := InvolutiveBasis(L, var);
                J := [-x + y3]
> FactorModuleBasis(var);
                 $\frac{1}{1-x} + \frac{y}{1-x} + \frac{y^2}{1-x}$ 
> PolHilbertSeries(t);
                 $1 + 2t + 3t^2 + 3\frac{t^3}{1-t}$ 

```

Now we work with the pure lexicographical ordering (note still $x > y$):

```

> J1:=InvolutiveBasis(L,var,1);
                                J1 := [x - y3]
> FactorModuleBasis(var);
                                 $\frac{1}{1-y}$ 
> PolHilbertSeries(t);
                                 $1 + \frac{t}{1-t}$ 

```

The grading is determined by the filtering according to the degrees of the monomials in the basis computed by `FactorModuleBasis`, which again results from the information contained in `PolTabVar`. There is yet another possibility to modify this grading, i. e. the choice of the leading terms. One can assign a natural number d_i as degree to the variable x_i and in the proper module case $q > 1$ a non negative integer c_i to the i -th standard R -basis vector $e_i := (0, \dots, 0, 1, 0, \dots, 0)$. The degree of the term $x_1^{\alpha_1} \cdots x_n^{\alpha_n} e_i$ is then $c_i + \sum \alpha_j d_j$ instead of $\sum \alpha_j$ in the old regime. When choosing the leading term, this degree is taken into account first, and, in case there are two terms of the same new degree, the old regime is applied for the final decision. If in doubt, one can use the function `LeadingMonomial`. Note a subtle point at this stage: The functions `FactorModuleBasis` and `PolHilbertSeries` start from the information contained in `PolTabVar` which does no longer know anything about these modified degrees. If one wants the true Hilbert series of the graded ring with the grading according to the new degrees, one has to use the function `PolWeightedHilbertSeries`, which takes the degrees as part of its input.

The final point in this section concerns speed. For some, but not all of the commands above, there are C++-versions available, which of course can be called from within MAPLE.

Command:	Fast Version:
<code>InvolutiveBasis</code>	<code>InvolutiveBasisFast</code>
<code>PolInvReduce</code>	<code>PolInvReduceFast</code>

For these commands up to now only the defaults work. Moreover, the C++-versions assume that the field of constants is the field of rational numbers, whereas the MAPLE-version can deal with extensions of the rationals, e. g. with purely transcendental extensions, which is often useful, cf. example 5.3. For big examples, the C++-versions are faster by a factor 1000. If one wants to see the HILBERT series or continue to work with some other of the MAPLE programs, the command `AssertInvBasis` produces the data contained in `PolTabVar`, and one can proceed.

4 Syzygies

Having discussed the cokernel $M := R^q/S$ of an R -module homomorphism $R^a \rightarrow R^q$ in the last two sections, the main issue will be the kernel of such a map here, i. e. the

relations between the generators of S . The idea for their computation is to introduce right hand sides. We redo Example 2.2 with right hand sides.

Example 4.1 (*Syzygies in Example 2.2*)

> `var:=[x,y];`

$$\text{var} := [x, y]$$

Introduce names for the generators of the module and take them as right hand sides:

> `L:=[[x,-y]=a,[y,x]=b,[y^3,0]=c];`

$$L := [[x, -y] = a, [y, x] = b, [y^3, 0] = c]$$

Now the Janet basis also has a right hand side, expressing the basis vectors as a combination of the original generators:

> `J:=InvolutiveBasis(L,var);`

$$J := [[y, x] = b, [x, -y] = a, [0, x^2 + y^2] = bx - ay, [0, y^2 x] = -c + by^2, [0, y^4] = cx - ay^3]$$

Now also a coset representative can be given a name and the standard representative will also be given together with its expression in terms of the original representative and the original generators of the submodule:

> `PolInvReduce([x^2,y^2]=u,J,var);`

$$[0, yx + y^2] = -ax + u$$

Finally the relations between the original generators can be computed:

> `Syzygies(L,var);`

$$[cx^2 - ay^3x + cy^2 - by^4]$$

If one iterates the computation of the syzygies, i. e. computes the second, third, and higher syzygies, one gets a free resolution. This can be done in one go by the command `PolResolution`, as the next example shows. However the resolution starts with the JANET basis and not with the original relations.

Example 4.2 (*Resolution in Example 2.2*)

> `var:=[x,y]:`

Enter generators without right hand side and compute the Janet basis:

> `L:=[[x,-y],[y,x],[y^3,0]]:`

> `J:=InvolutiveBasis(L,var);`

$$J := [[y, x], [x, -y], [0, x^2 + y^2], [0, y^2 x], [0, y^4]]$$

The following command computes a free resolution of the residue class module M (even without invoking the last command first), where, however, one does not work with the original generators of S but with the Janet basis, cf. columns of the first matrix. The columns of the second matrix give the relations between the columns of the first matrix. In general the procedure stops after k steps, where k is the maximum number of nonmultiplicative variables.

> `PolResolution(L,var);`

$$\left[\begin{array}{ccccc} y & x & 0 & 0 & 0 \\ x & -y & x^2 + y^2 & y^2 x & y^4 \end{array} \right], \left[\begin{array}{ccc} 0 & 0 & x \\ 0 & 0 & -y \\ 0 & -y^2 & -1 \\ -y^2 & x & 0 \\ x & 1 & 0 \end{array} \right]$$

5 Further Examples

In this section some examples demonstrating the use of the JANET algorithm will be presented. The first example concerns the **intersection** of submodules of R^q . We employ the well known technique by ZASSENHAUS: If $S_1, S_2 \leq R^q$, then $S_1 \cap S_2$ is the kernel of the projection

$$R^{2q} \rightarrow R^q : (a, b) \mapsto a$$

restricted to $\langle (s_1, s_1), (s_2, 0) \mid s_1 \in S_1, s_2 \in S_2 \rangle$. The default ordering in the module case provides the JANET basis for the intersection automatically as the subsequent example demonstrates in the case $q = 1$.

Example 5.1 *The intersection of the ideals generated by L1 and L2 is computed as follows:*

```
> var:=[x,y]:
> L1:=[x^6-y^6]:
> L2:=[x^9-y^9]:
> l12:=map(a->[a,a],L1):
> l21:=map(a->[a,0],L2):
> L:=[op(l12),op(l21)];
      L := [[x^6 - y^6, x^6 - y^6], [x^9 - y^9, 0]]
> J:=InvolutiveBasis(L,var);
      J := [[x^6 - y^6, x^6 - y^6], [-y^9 + y^6 x^3, -x^9 + y^6 x^3], [-y^9 x + y^6 x^4, -x^10 + y^6 x^4],
[-y^9 x^2 + y^6 x^5, -x^11 + y^6 x^5], [0, x^12 - y^12 + x^9 y^3 - y^9 x^3]]
> N:=map(a->if a[1]=0 then a[2] fi,J);
      N := [x^12 - y^12 + x^9 y^3 - y^9 x^3]
N is already a Janet basis for the intersection. Here a check:
> JN:=InvolutiveBasis(N,var);
      JN := [x^12 - y^12 + x^9 y^3 - y^9 x^3]
```

Up to this point everything works also for nonprincipal ideals. Since in the present case the two ideals to be intersected were principal, $JN[1]$ is the least common multiple of $L1[1]$ and $L2[1]$. Here is the greatest common divisor:

> *simplify*($L1[1]*L2[1]/JN[1]$);
 $x^3 - y^3$

The next example concerns the computation of the **annihilator** of the module $M = R^q/S$, i. e. the ideal of all $a \in R$ with $aM = 0$. Obviously this is the intersection of the q ideals $I_i \trianglelefteq R$ such that

$$\underbrace{\{0\} \oplus \cdots \oplus \{0\}}_{i-1} \oplus I_i \underbrace{\{0\} \oplus \cdots \oplus \{0\}}_{q-i} \leq S$$

with $i = 1, \dots, q$. Note, the default version (position over term) of the command `InvolutiveBasis` yields a JANET basis for I_q immediately. Since the procedure for taking intersections of ideals was already demonstrated above, the following example is for a module invariant under the cyclic permutation of the positions so that `InvolutiveBasis` yields the JANET basis for the intersection in one go.

Example 5.2 *Computation of the annihilator of R^3/S where S is generated by L below.*

> *var*:= $[x, y, z]$;
> *L*:= $[[x^2, y, z], [y, z, x^2], [z, x^2, y], [x+y, x+y, x+y]]$;
> *J*:=*InvolutiveBasis*(*L*, *var*);

J := $[[z, x^2, y], [y, z, x^2], [x, x + y - z, x + y - x^2],$
 $[0, -xz + xy + y^2 - zy, -x^3 + xy + y^2 - x^2y],$
 $[0, x^2 - y^2 + zy - y, -z + x^2 - y^2 + x^2y], [0, -z^3 + z^2y - zy + z^2,$
 $-x^4y + zy^2 - x^2z^2 - y^2 + 2x^2z - x^4 + x^2y - y^3 + x^2y^2 - z^2 + zy^2],$
 $[0, zy^2 - z^3 - y^2 + z^2, -zy + x^2z + 2zy^2 - x^4y - x^2z^2 - y^2 + x^2y - y^3 + x^2y^2],$
 $[0, y^3 - z^3, -x^4y + 2zy^2 - x^2z^2],$
 $[0, 0, x^5 - zyx + z^2x - x^3z + x^4y - zy^2 + z^2y - zyx^2 - x^3y + y^2x + y^3 - x^2y^2], [$
 $0, 0, x^4y - y^3z + z^2y^2 + x^4z - 2zyx^2 - x^2y^2 - x^2y^2z + y^3 - x^2y^3 + y^4 + z^3 - x^2z^2$
 $+ x^4y^2]]$
> *N*:=*map*(*a*->*if* (*a*[1]=0 and *a*[2]=0) then *a*[3] *fi*, *J*);

N := $[x^5 - zyx + z^2x - x^3z + x^4y - zy^2 + z^2y - zyx^2 - x^3y + y^2x + y^3 - x^2y^2, x^4y - y^3z$
 $+ z^2y^2 + x^4z - 2zyx^2 - x^2y^2 - x^2y^2z + y^3 - x^2y^3 + y^4 + z^3 - x^2z^2 + x^4y^2]$

Since L and hence also S is closed under cyclic permutation of the three components, N is a Janet basis for the annihilator of S . (The first and second components give the same ideal as the third, which was picked here.) Here is a confirmation for the first component:

> *map*(*n*->*PolInvReduce*($[n, 0, 0]$, *J*, *var*), *N*);
 $[[0, 0, 0], [0, 0, 0]]$

The next example demonstrates an unexpected use of the JANET algorithm, which is designed as an algorithm for modules. By a trick it is also able to produce ring relations.

Example 5.3 Obviously the three polynomials x^2+y^2 , x^2*y^2 , x^3*y-y^3*x are algebraically dependent. To find a relation between them three new variables a,b,c are introduced and the following computation is carried out over the field of rational functions in a,b,c over rationals:

```
> var:=[x,y];
                                var := [x, y]
> L:=[x^2+y^2-a,x^2*y^2-b,x^3*y-y^3*x-c];
                                L := [x^2 + y^2 - a, x^2 y^2 - b, x^3 y - y^3 x - c]
> J:=InvolutiveBasis(L,var);
                                J := [-a c^3 + b a^3 c - 4 b^2 a c]
```

This result means, that L generates all of R . It can also be interpreted as a relation between the three original polynomials.

The conventional way to proceed in the last example would have been to have a,b,c as additional variables and to work with 5 variables over the rationals using the elimination order to get rid of x and y . This is more time consuming, but in principle working always. Whereas the above trick only gives a first relation.

Example 5.4 Conventional method to redo the previous example:

```
> var:=[x,y,a,b,c];
                                var := [x, y, a, b, c]
> L:=[x^2+y^2-a,x^2*y^2-b,x^3*y-y^3*x-c];
> J:=InvolutiveBasis(L,var,1);
                                J := [-c^2 + b a^2 - 4 b^2, -4 b^2 y - c^2 y + a^2 b y, y^2 a^2 b - c^2 y^2 - 4 y^2 b^2,
                                y^3 a^2 b - c^2 y^3 - 4 y^3 b^2, b + y^4 - a y^2, c x + 2 b y + a y^3 - a^2 y,
                                a c x + y^3 a^2 - a^3 y + 2 b a y, a^2 c x + y^3 a^3 - a^4 y + 8 b^2 y + 2 c^2 y,
                                y^3 c a - a^2 c y + a^2 b x - 4 b^2 x + 2 c y b, c y x + 2 b y^2 - a b,
                                a c x y + 2 b a y^2 - 4 b^2 - c^2, a^2 y x - 4 b y x + 2 c y^2 - a c, c x y^2 + 2 b y^3 - b a y,
                                2 b y^2 x - y^3 c - a b x + a c y, -2 b x + a y^2 x + c y, 2 y^3 x + c - a y x, x^2 + y^2 - a]
> NotHas(J,var,[x,y]);
                                [-c^2 + b a^2 - 4 b^2]
```

Note, this result differs by a factor $a*c$ from the result obtained in the previous example.

Further examples of the package JANET may be found in [CPI 01].

6 Acknowledgements

The contribution of two authors (Yu.A.B. and V.P.G.) was partially supported by the grants 00-15-96691 and 01-01-00708 from the Russian Foundation for Basis Research.

References

- [AGW 03] S. Abenda, G. Gaeta, S. Walcher (eds.), *Symmetry and Perturbation Theory SPT2002*. to appear World Scientific 2003.
- [Ape 98] J. Apel, *The Theory of Involutive Divisions and Applications to Hilbert Function Computations*. *J. Symbolic Computation* (1998) **25**, 683-704.
- [BGe 01] Y. A. Blinkov, V. P. Gerdt, D. A. Yanovich, *Construction of Janet bases II. Polynomial Bases*. 249-263 in [GMV 01].
- [CPl 01] C. F. Cid, W. Plesken, *Invariants of finite groups and involutive division*. 122-135 in [GMV 01].
- [GMV 01] V. G. Ganzha, E. W. Mayr, E. V. Vorozhtsov (eds.) *Computer Algebra in Scientific Computing CASC 2001*. Springer Berlin etc. 2001.
- [GeB 98a] V. P. Gerdt, Y. A. Blinkov, *Involutive bases of polynomial ideals*. *Mathem. and Computers in Simulation* 45 (1998), 519-541.
- [GeB 98b] V. P. Gerdt, Y. A. Blinkov, *Minimal involutive bases*. *Mathem. and Computers in Simulation* 45 (1998), 543-560.
- [Ge 99] V. P. Gerdt, *Completion of Linear Differential Systems to Involution*, In: *Computer Algebra in Scientific Computing / CASC'99*, V.G.Ganzha, E.W.Mayr, E.V.Vorozhtsov (eds.), Springer-Verlag, Berlin (1999) 115-137.
- [Ge 02] V. P. Gerdt, *Involutive Division Technique: Some Generalizations and Optimizations*. *Journal of Mathematical Sciences* 108(6), 2002, 1034-1051.
- [Jan 29] M. Janet, *Lecons sur les systèmes des équations aux dérivées partielles*. Cahiers Scientifique IV, Gauthiers-Villars, Paris 1929.
- [Ple 02] W. Plesken, *JANETS Algorithm*. to appear in [AGW 03].
- [Pom 94] J.- F. Pommaret, *Partial Differential Equations and Group Theory*. Kluwer Academic Publishers 1994.